

MYSTERIOUS

ADVENTURES

FOR YOUR AMSTRAD

By Tim Hartnell
and Clive Gifford



**MYSTERIOUS
ADVENTURES
FOR YOUR
AMSTRAD**

MYSTERIOUS ADVENTURES FOR YOUR AMSTRAD

**By
Clive Gifford
and
Tim Hartnell**

*Best Wishes,
C. Gifford
Virgin*

First published in Great Britain in 1985 by Virgin Books Ltd,
328 Kensal Road, London W10 5XJ.

Copyright © 1985 Interface/Virgin Books

ISBN 0 86369 116 1

All rights reserved. No part of this book may be reproduced in any
form or by any means without prior permission from the publisher.

Printed and bound in Great Britain by Richard Clay
(The Chaucer Press) Ltd, Suffolk.

Designed and illustrated by Sue Walliker.

Typeset by Keyline Graphics.

Distributed by Arrow Books.

The Authors

Tim Hartnell is one of the most widely-published computer authors in the world. His books include *The Personal Computer Handbook* (Virgin Books, 1984), *Creating Adventure Programs on Your Microcomputer* (Ballantine, US, 1984) and several other works in the Virgin Games series. He is managing director of the specialist publishing house, Interface Publications Ltd., and writes the regular column "Response Frame" for *Your Computer* magazine, as well as contributing to several other computer publications.

Nineteen-year-old Clive Gifford has had nearly 20 computer books published in the last three years. He is editor of this series of adventure books from Virgin. He is author of *The Amstrad Pentacle Adventure Creator* (Interface, 1985), a 64-page book containing an adventure-generating program. Clive also wrote *Dynamic Games for Your Amstrad* (Interface, 1984) and the widely-acclaimed *Using Computers in Education*.

CONTENTS

Introduction	9
The Explorer	11
The Patriarch	17
3-D Maze/The Memnonite Stone	31
Into the Unknown	43
Gotham Runner	54
Golden Nugget	64
Writing Your Own Adventures	85
Bibliography	93

Introduction

The world of adventures and adventuring is a rapidly growing one. The popularity of such games has rocketed from little more than a cult following to a major sector of the games market.

You can now explore this world with your Amstrad and this book. We've tried to include as many different types of adventure as possible in this book, to give you a wide range of examples to enjoy. When you come to writing your own adventures (and we've included some solid advice at the back of the book to help you with this), the experience gained from playing through the adventures in this book will ensure you'll be able to choose from a number of different types to create the kind of adventure which suits you best.

All in all, you and your Amstrad are set for a busy and entertaining couple of months ahead!

Tim Hartnell,
Clive Gifford,
London, September 1985

N.B. One technical point: where you see a '£' sign that is *not* within speechmarks, change it to a '#' (Hash-SHIFT and '3').

THE EXPLORER

We'll start the book with a small and fairly simple adventure, just to get you on the way. In this adventure, you're exploring the environment around you, looking for the Princess. If the choice of object for your quest seems a little sexist, by all means change it into a Prince, a toadstool, or even your favourite television presenter (see line 910).

The commands are entered in the well-known "verb noun" format, so you should try things like GO NORTH, CHOP TREE and CLIMB ROPE. You'll soon discover which words work with this adventure.

By the way, we suggest that with *The Explorer* – as with most of the other adventures in this book – you make a map of the adventure environment as you are exploring it.



```

10 REM*****
20 REM*****THE EXPLORER*****
30 REM**A CLASSIC MINI-ADVENTURE**
40 REM**FOR BEGINNER ADVENTURERS**
50 REM*****
60 GOSUB 880
70 INK 3,6:INK 1,24:BORDER INT(RND*26)
80 CLS:L=2
90 GOTO 740
100 PRINT:PRINT">>>>>>>>>>>>>>>>>>>>>>>>>"
110 PEN 2
120 PRINT:PRINT
130 Z=Z+1:IF Z=2 THEN Z=1:PEN 3 ELSE PEN
    2
140 SOUND 1,INT(RND*400)+30
150 BORDER INT(RND*26)
160 PRINT "I AM ON":PRINT L$(L)
170 PRINT "AND I AM CARRYING:";X=X+1:GOSU
    B 690
180 PRINT "I CAN SEE:";X=L:GOSUB 690
190 PRINT:INPUT "WHAT NEXT";A$
200 A$=UPPER$(A$)
210 K$=A$
220 IF A$="" THEN 100
230 V$=""
240 J=LEN(A$):FOR M=1 TO J
250 IF LEFT$(A$,1)=" " THEN GOSUB 680:GO
    TO 280:GOTO 680:GOTO 280
260 GOSUB 670
270 NEXT M

```

```
280 IF V$="GET" THEN 350
290 IF V$="DROP" THEN 450
300 IF V$="CHOP" THEN 520
310 IF V$="GO" THEN 560
320 IF V$="HELP" THEN GOTO 650
330 PRINT "I CANNOT ";K$
340 GOTO 100
350 X=0
360 FOR B=0 TO 9
370 IF BL(B)=L AND A$=B$(B) THEN X=B:BL(
B)=0
380 NEXT B
  70 IF X=0 THEN PRINT "IT IS NOT HERE.":
GOTO 100
400 IF X=5 AND BL(6)=4 THEN BL(5)=4:PRIN
T "THE WITCH STOPS ME.":GOTO 100
410 IF X=1 OR X=9 THEN PRINT "DON'T BE S
ILLY!":BL(X)=L
420 IF X=4 AND L=2 AND BL(3)<0 THEN BL(3
)=2:PEN 1:PRINT"££££ I SEE GOLD ££££":PE
N 3
430 IF X<>5 THEN 100
440 PRINT "CONGRATULATIONS, YOU HAVE WON
THE      PRINCESS!!!":SOUND 1,100,100:
END
450 X=0
460 FOR B=1 TO 9
470 IF BL(B)=0 AND A$=B$(B) THEN X=B:BL(
B)=L
480 NEXT B
```

```

490 IF X=2 AND L=2 THEN BL(2)=1:BL(8)=2
500 IF X=0 THEN PRINT "I DO NOT HAVE IT.
"
510 GOTO 100
520 IF A$="TREE" AND L=1 AND BL(7)=0 AND
    BL(1)=1 THEN DV=1
530 IF DV=1 THEN PRINT "TIMBERRRR!":BL(1
)=1:BL(2)=1:GOTO 100
540 IF A$="TREE" THEN PRINT "WITH WHAT?"
:GOTO100
550 GOTO 330
560 IF A$="NORTH" THEN X=0
570 IF A$="SOUTH" THEN X=1:GOTO 630
580 IF A$="EAST" THEN X=2:GOTO 630
590 IF A$="WEST" THEN X=3:GOTO 630
600 IF L=2 AND A$="LOG" AND BL(8)=2 THEN
    L=3:GOTO 100
610 IF L=3 AND A$="SHORE" THEN L=4:LL(L,
X)=4
620 IF L=4 AND BL(3)=0 THEN BL(3)=-1:BL(
6)=-1: PRINT "THE WITCH TOOK THE GOLD AN
D RAN!":GOTO 100
630 IF LL(L,X)<0,THEN PRINT N$(-LL(L,X))
:GOTO 100
640 L=LL(L,X):GOTO 100
650 IF L=2 AND (BL(7)=0 OR BL(7)=2) THEN
    PRINT "DROP LOG IN WATER...THEN 'GO LOG
"
660 GOTO 100
670 V$=V$+LEFT$(A$,1)

```

```
680 A$=RIGHT$(A$,LEN(A$)-1):RETURN
690 FOR B=1 TO 9
700 IF BL(B)=X THEN PRINT " ", " ";B$(B);
710 NEXT B
720 PRINT
730 RETURN
740 DIM L$(4),LL(4,3),B$(9),BL(9),N$(2)
750 DATA "THE EDGE OF THE FOREST","THE N
ORTH EDGE OF THE LAKE"
760 DATA "A LOG WHICH I PADDED ACROSS T
HE LAKE"
770 DATA"THE SOUTH SIDE OF A LAKE"
780 DATA -2,2,-2,-2,1,-1,-1,-1,-1,-1,
-1,-1,-2,-2,-2
790 DATA TREE,LOG,GOLD,ROCK,PRINCESS,WIT
CH,AXE,LOG IN THE WATER,SHORE
800 DATA 1,-1,-1,2,4,4,0,-1,3
810 DATA I CANNOT SWIM, THERE IS NOTHING
THAT WAY
820 FOR I=1 TO 4:READ L$(I):NEXT I
830 FOR I=1 TO 4:FOR J=0 TO 3:READ LL(I,
J):NEXT J:NEXT I
840 FOR I=1 TO 9:READ B$(I):NEXT
850 FOR I=1 TO 9:READ BL(I):NEXT
860 READ N$(1):READ N$(2)
870 GOTO 100
880 REM***INSTRUCTIONS***
890 CLS
900 LOCATE 4,2:PRINT "This mini-adventur
e involves you"
```



```

910 PRINT "exploring the area around you
    to find a way to win the Princess of Li
ght...who,"
920 PRINT "if rumours are to be believed
    , is      somewhere nearby."
930 PRINT:PRINT
940 PRINT "      To control your characte
r in this  game requires you to enter th
e command  and the object it is to affec
t."
950 PRINT "For example, 'GET KNIFE'. For
    movement, the direction (based on the c
ompass      points) must be preceded by t
he 'GO'      command."
960 PRINT:PRINT
970 PRINT "      Your final help is a lis
t of      commands."
980 PRINT "GO NORTH; GO SOUTH; GO EAST;
GO WEST; GET; DROP; CHOP; HELP."
990 PRINT:PRINT:PRINT
1000 WHILE INKEY$="":WEND
1010 RETURN

```


THE PATRIARCH

Many adventures are constructed on a "grid system", although the actual grid is rarely presented explicitly to the player. In this adventure, things are different. You are exploring a cave system, and it is built on a grid.

Run in **MODE 0** – to take advantage of the large number of colours that can be used in this mode – *The Patriarch* places you in a grid-system world populated by fierce dragons, quicksand pits, magic caves, and, of course, treasure. As many adventure clichés as possible have been squeezed into this world.

When you first run the program, you will be told which user-defined graphic represents which item in the world. (You're sure to be very flattered with the smiling face which is your user-defined image!)


Once you've been told which user-defined graphic is which, you'll be given a (very) brief glimpse at the world from above. Look quickly, and try to remember where all the elements of your world are located, so you can head for the gold, while avoiding the quicksand...



However, it's not all just up to your memory. You own a "magic amulet" (another cliché which fell out of somebody else's adventure), and this is used to indicate what is in the caves around you. But because you're not much of a whiz when it comes to envoking magic, the old amulet doesn't work too well. Certainly you'll know what is in at least one of the caves near you, but you won't know which cave the amulet information refers to.

You'll find it easy to play this adventure (although it is not so easy to survive) as most of your actions are chosen from a menu of possibilities that is presented after each move. Because the adventure runs on a very early model Amstrad, you'll find the text output resembles that from an ancient (but colourful) telex machine (see the subroutine from line 2500, if you want to include this type of output in your own programs).

And now, Patriarch, it is time to begin, to battle a few of those fierce dragons, search for treasure, and all that!



```
10 REM The Patriarch
20 GOSUB 1560
30 CLS:PRINT:PRINT:PRINT
40 PEN 1
50 GOSUB 1440
60 Q=INT(RND*7)
70 IF Q=0 AND E<>55 THEN GOSUB 1440
80 PEN 2
90 CLS:PRINT:PRINT:PRINT "Patriarch ";A$
:",";ME=1:GOSUB 2500:PRINT E
100 IF G>0 THEN ME=2:PEN 2:GOSUB 2500:PR
INT G
110 GOSUB 970
120 PEN 3
130 IF H<25 THEN 170
140 PRINT "THE GAME IS OVER, ASALL YOUR
CHARISMA HAS BEEN EXHAUSTED!"
150 FOR Z=1 TO 4000:NEXT Z:PRINT:PRINT "
BUT YOU PLAYED WELL,PATRIARCH ";A$
160 PRINT:FOR Z=1 TO 4000:NEXT Z:IF G>0
THEN PRINT "AND YOU MANAGED TO AQUIRE T
REASURE WORTH $"G:PRINT:Q=9:GOTO 143
0
170 PRINT:ME=3:GOSUB 2500:PRINT 25-H
180 PEN 6
190 PRINT:PRINT "WHAT DO YOU WANT TO DO?
"
200 WHILE INKEY$<>"":WEND
210 PEN 4
```

```

220 PRINT "N - move North      S - move
South"
230 PRINT "E - move East      W - move
West"
240 PRINT "F - Fight dragon    Q - Quit"
250 Z$=""
260 WHILE Z$<>"N" AND Z$<>"S" AND Z$<>"E
" AND Z$<>"W" AND Z$<>"F" AND Z$<>"Q":Z$
=INKEY$:Z$=UPPER$(Z$):WEND
270 IF Z$="N" AND A(E-10)=240 OR Z$="S"
AND A(E+10)=240 OR Z$="E" AND A(E+1)=240
OR Z$="W" AND A(E-1)=240 THEN PRINT:SOU
ND 1,220,99,7,0,0,1:ME=4:GOSUB 2500:GOTO
250
280 SOUND 2,20,7:SOUND 2,50,7
290 INK 3,26:PEN 3
300 IF Z$="Q" THEN Q=9:GOTO 1410
310 A(E)=245
320 IF Z$="N" THEN E=E-10
330 IF Z$="S" THEN E=E+10
340 IF Z$="E" THEN E=E+1
350 IF Z$="W" THEN E=E-1
360 IF Z$="F" THEN GOSUB 1150
370 IF A(E)=241 THEN PEN 1:GOSUB 450:REM
MAGIC
380 IF A(E)=242 THEN PEN 6:GOSUB 550:REM
DRAGON
390 IF A(E)=243 THEN PEN 4:GOSUB 710:REM
QUICKSAND
400 IF A(E)=244 THEN PEN 5:GOSUB 830

```




```
410 H=H+1:IF F=25 THEN Q=9:GOTO 1400
420 GOSUB 1980
430 GOTO 60
440 REM -----
450 REM MAGIC
460 PRINT:ME=5:GOSUB 2500
470 GOSUB 1980
480 A(E)=245
490 FOR J=2 TO 800 STEP 50
500 SOUND 5,J,7:SOUND 2,1000-J,7
510 NEXT J
520 E=INT(RND*76)+12:WHILE A(E)=240:GOTO
520:WEND
530 RETURN
540 REM -----
550 REM DRAGON
560 ME=6:GOSUB 2500
570 GOSUB 1980
580 M=RND
590 IF M<0.2 THEN ME=7:GOSUB 2500:RETURN
600 ME=8:GOSUB 2500
610 GOSUB 1980
620 IF M>0.85 THEN ME=9:RETURN
630 ME=10:GOSUB 2500
640 GOSUB 1980
650 M=RND
660 IF M>0.95 OR H>18 THEN ME=11:GOSUB 2
500:RETURN
670 ME=12:GOSUB 2500:PRINT " ";A$
680 GOSUB 1980
```

```

690 Q=9:GOTO 1400
700 REM -----
710 REM QUICKSAND
720 FOR J=1 TO 12
730 PEN INT(RND*6+1)
740 PRINT TAB(J/2);"HORRORS"
750 PRINT
760 PRINT TAB(10-J);"QUICKSAND!"
770 SOUND 5,CINT(RND*7+1000),7
780 PRINT
790 FOR DELAY=1 TO 90:NEXT DELAY
800 NEXT J
810 Q=9:H=0:GOTO 1400
820 REM -----
830 REM TREASURE
840 K=INT(RND*100)+100
850 FOR J=1 TO 13
860 PEN INT(RND*7+1)

```



```
870 PRINT TAB(J/2); "TREASURE!":PRINT
880 SOUND 5,CINT(RND*70+10),7
890 PRINT TAB(10-J/2); "TREASURE":PRINT
900 FOR DELAY=1 TO 200:NEXT DELAY
910 NEXT J
920 GOSUB 1980
930 PRINT:ME=13:GOSUB 2500:PRINT K
940 G=G+K
950 RETURN
960 REM -----
970 REM AMULET
980 PEN 1
990 Y=1
1000 L=A(E+P(Y))
1010 IF L<>245 THEN 1050
1020 IF Y<8 THEN Y=Y+1:GOTO 1000
1030 IF L=245 THEN RETURN
1040 PRINT "PATRIARCH ";A$;","
1050 ME=14:GOSUB 2500
1060 IF L=240 THEN ME=15:GOSUB 2500
1070 IF L=241 THEN ME=16:GOSUB 2500
1080 IF L=242 THEN ME=17:GOSUB 2500
1090 IF L=243 THEN ME=18:GOSUB 2500
1100 IF L=244 THEN ME=19:GOSUB 2500
1110 PRINT " ";:ME=20:GOSUB 2500
1120 GOSUB 1980
1130 RETURN
1140 REM -----
1150 REM ATTACK DRAGON
1160 PRINT
```

```

1170 AR=AR-1
1180 IF AR=0 THEN ME=21:GOSUB 2500:GOSUB
1980:RETURN
1190 ME=22:PRINT ar;:GOSUB 2500
1200 SS=0
1210 ME=23:GOSUB 2500
1220 Z$=INKEY$:Z$=UPPER$(Z$)
1230 WHILE Z$<>"N" AND Z$<>"S" AND Z$<>"
E" AND Z$<>"W":GOTO 1220:WEND
1240 IF Z$="N" AND A(E-10)=242 THEN SS=1
:YT=E-10
1250 IF Z$="S" AND A(E+10)=242 THEN SS=1
:YT=E+10
1260 IF Z$="E" AND A(E+1)=242 THEN SS=1:
YT=E+1
1270 IF Z$="W" AND A(E-1)=242 THEN SS=1:
YT=E-1
1280 PRINT
1290 IF SS=0 THEN ME=24:GOSUB 2500:GOTO
1390
1300 ME=25:GOSUB 2500
1310 GOSUB 1980
1320 IF RND>0.3 THEN 1380
1330 ME=26:GOSUB 2500
1340 A(YT)=245
1350 K=INT(RND*100)+100
1360 PRINT
1370 ME=27:GOSUB 2500:PRINT K:G=G+K:GOTO
1390
1380 ME=28:GOSUB 2500


```

```
1390 GOSUB 1980:RETURN
1400 IF H<1 THEN ME=29:GOSUB 2500:GOSUB
1980:GOTO 1420
1410 ME=30:GOSUB 2500:PRINT 25-H
1420 ME=31:IF G>0 THEN GOSUB 2500:PRINT
G
1430 PRINT:PRINT
1440 A(E)=246
1450 FOR J=1 TO 100
1460 X=A(J)-239
1470 PEN X
1480 SOUND 5,A(J)*2,7
1490 PRINT CHR$(A(J));" ";
1500 IF 10*INT(J/10)=J THEN PRINT
1510 NEXT J
1520 GOSUB 1980
1530 IF Q=9 THEN END
1540 RETURN
1550 REM -----
1560 MODE 0:BORDER 0:PRINT:PRINT:PRINT
1570 INK 0,0:INK 1,2:INK 2,6:INK 3,8:INK
4,18:INK 5,24:INK 6,26:INK 7,15
1580 PEN 3
1590 PRINT "Welcome...":PRINT:PRINT
1600 PEN 6
1610 INPUT "What is your name, Patriar
ch";A$
1620 A$=UPPER$(LEFT$(A$,1))+LOWER$(MID$(
A$,2))
1630 CLS:PRINT:PRINT:PEN 4
```

```

1640 PRINT "Please stand by      Patriarc
h ";A$
1650 DIM A(100),Q$(31)
1660 H=0:Q=0:L=0:G=0:AR=6
1670 FOR B=1 TO 100
1680 A(B)=245
1690 IF B<12 OR B>90 OR 10*INT(B/10)=B O
R 10*INT(B/10)=B-1 THEN A(B)=240
1700 NEXT B
1710 FOR B=1 TO 5:RESTORE
1720 FOR D=1 TO 5
1730 Z=INT(RND*76)+12:IF A(Z)=240 THEN 1
730
1740 READ C:A(Z)=C
1750 NEXT:NEXT
1760 DATA 240,241,242,243,244
1770 FOR B=1 TO 8:READ Z:P(B)=Z:NEXT B
1780 DATA -11,-10,-9,-1,1,9,10,11
1790 E=55
1800 RESTORE 2010
1810 FOR J=1 TO 31:REM READ MESSAGES
1820 READ M$:Q$(J)=M$
1830 NEXT J
1840 GOSUB 2330:REM DEFINE CHARACTERS
1850 CLS
1860 PRINT:PRINT:PRINT "KEY TO SYMBOLS..
.":PRINT
1870 PEN 1:PRINT:PRINT "   ";CHR$(240);"
WALL ";CHR$(240)
1880 PEN 2:PRINT:PRINT "   ";CHR$(241);"

```



```

MAGIC CAVE ";CHR$(241)
1890 PEN 3:PRINT:PRINT " ";CHR$(242);"
DRAGON ";CHR$(242)
1900 PEN 4:PRINT:PRINT " ";CHR$(243);"
QUICKSAND ";CHR$(243)
1910 PEN 5:PRINT:PRINT " ";CHR$(244);"
GOLD ";CHR$(244)
1920 PEN 6:P$=UPPER$(A$):PRINT:PRINT "
";CHR$(246);" ";P$;" ";CHR$(246)
1930 PEN 7:PRINT:PRINT:PRINT " Press an
y key to continue..."
1940 WHILE INKEY$="":LL=RND:WEND
1950 RETURN
1960 REM -----
1970 REM DELAY
1980 FOR DELAY=1 TO 1000:NEXT DELAY
1990 RETURN
2000 REM -----

```

2010 DATA You are in cave
 2020 DATA Treasure \$
 2030 DATA CHARISMA LEFT
 2040 DATA YOU CAN'T MOVE THAT WAY!!
 2050 DATA YOU'VE STUMBLERD INTO A MAG
 IC CAVE & WILL NOW BE WHISKED TO ANOTHER
 !
 2060 DATA YOU'VE WANDERED INTO A DRA
 GON'S LAIR! START SAYING YOUR PRAYE
 RS!
 2070 DATA IT HAS FLOWN AWAY
 2080 DATA IT AWAKENS AND HAS SEEN YOU!
 2090 DATA BUT IT HAS RECENTLY EATEN AND
 SO GOES BACK TO SLEEP
 2100 DATA AND NOW IT ATTACKS!
 2110 DATA BUT YOU FIGHT BACK AND WIN!
 2120 DATA GOODBYE PATRIACH
 2130 DATA YOU'VE FOUND A HOARD OF D
 RAGON'S GOLD WORTH \$
 2140 DATA # YOUR AMULET SIGNAL
 S THERE IS
 2150 DATA A SOLID WALL
 2160 DATA A MAGIC CAVE
 2170 DATA A DRAGON
 2180 DATA QUICKSAND
 2190 DATA GOLD
 2200 DATA NEARBY
 2210 DATA YOU HAVE USED UP ALLYOUR ARROW
 S...
 2220 DATA ARROWS LEFT

2230 DATA WHICH DIRECTION DO YOU WANT T
O SHOOT IN (N S E or W)?

2240 DATA THERE WAS NO DRAGON THERE SO Y
OU HAVE WASTED AN ARROW

2250 DATA WELL DONE PATRIARCH. YOU'VE HI
T A FEROCIOUS DRAGON!

2260 DATA YOU KILLED IT!!

2270 DATA YOU ARE REWARDED WITH \$

2280 DATA BUT YOU ONLY WOUNDED IT
!

2290 DATA ALL YOUR CHARISMA ISEXHAUSTED.

.....
.....
2300 DATA CHARISMA LEFT:

2310 DATA GOLD FOUND: \$

2320 REM -----

2330 REM DEFINE CHARACTERS

2340 REM SOLID WALL - 240

2350 SYMBOL 240,255,0,255,0,255,0,255,0

2360 REM MAGIC CAVE - 241

2370 SYMBOL 241,189,153,165,195,195,165,
153,126

2380 REM DRAGON - 242

2390 SYMBOL 242,24,126,153,255,126,126,2
4,126

2400 REM QUICKSAND - 243

2410 SYMBOL 243,255,182,109,219,182,109,
219,255

2420 REM GOLD - 244

```

2430 SYMBOL 244,16,56,124,254,254,124,56
,16
2440 REM ORDINARY WALL - 245
2450 SYMBOL 245,0,66,0,0,0,0,66,0
2460 REM PATRIARCH - 246
2470 SYMBOL 246,30,63,47,127,55,7,62,30
2480 RETURN
2490 REM -----
2500 REM 'TELETYPE' OUTPUT
2510 F$=Q$(ME)
2520 FOR J=1 TO LEN(F$)
2530 SOUND 5,CINT(RND*100+600),7
2540 PRINT MID$(F$,J,1);
2550 IF MID$(F$,J,1)=" " OR ME=3 THEN 25
80
2560 FOR P=1 TO 60:NEXT P
2570 IF ME=29 THEN PEN INT(RND*7+1)
2580 NEXT J
2590 IF ME<4 OR ME>11 AND ME<20 OR ME=24
OR ME=27 OR ME>29 THEN RETURN
2600 PRINT:PRINT
2610 RETURN

```

3-D Maze/The Memnonite Stone

On the morning of 27 December, 1666, Johann-Friedrich Schweitzer, a man described in the book *The Morning of the Magicians* (Louis Pauwels/Jacques Bergier) as "a violent anti-chemist" was visited by a stranger, dressed in a broad cloak, such as members of the Memnonite religious sect wore. The stranger presented Schweitzer with a fragment of rock, supposedly the "Philosopher's Stone", which had the power of changing base metals into gold.

We are not told whether the Stone ever worked, and the secret of its whereabouts died with Schweitzer.

Fortunately for mankind, the Stone has been detected, hidden within a maze. You have been chosen as the person to find it, in this 3-D maze adventure. Armed only with your Amstrad, and a patented *Memnonite Indicator*, you set out to brave the maze to locate the Stone.



The maze is modified from run to run of the program, and the Stone is placed at random within the maze. On the screen you'll see a picture of the maze ahead of you, projected on to the screen of a hitherto-unknown invention, the *Maze-O-Scope* (created with the aid of the Amstrad's WINDOW command).

Your goal, of course, is to find the Stone, and to do it in the smallest number of moves. You move through the maze using the cursor keys. Pressing the 'H' will summon up the "Help" option, which shows you the maze from above. You are shown on this maze as a white smiling face, and the path you've taken so far in the run is shown as a series of green dots. There is, however, a penalty for calling for help: the number of moves you've made so far in the run will be increased by five.

Don't rush through the maze when you first run this program. Wait until you become familiar with the view on your Maze-O-Scope, and you'll soon be surprised at how good you'll get at making your way around it. In addition, you will become very skilled at interpreting the output of your Memnonite Indicator, so you can significantly improve your time on subsequent runs.

And now here's the listing, so you can seek the Stone.



```

10 REM 3-D MAZE/THE MEMNONITE STONE
20 GOSUB 740
30 GOSUB 500
40 REM -----
50 WHILE NOT SOLVED
60 M=M+1
70 N=0
80 LOCATE 1,10
90 PRINT "MOVE NUMBER";:PEN 2:PRINT M:PE
N 1:PRINT
100 IF A$<>"H" THEN PRINT "You are facin
g ";
110 IF A$="N" THEN PRINT "North":Q$="Nor
th"
120 IF A$="S" THEN PRINT "South":Q$="Sou
th"
130 IF A$="E" THEN PRINT "East ":Q$="Eas
t"
140 IF A$="W" THEN PRINT "West ":Q$="Wes
t"
150 LOCATE 1,14
160 PRINT "NORTH: ";
170 IF A(D+1,E)=S THEN PRINT "OPEN":IF A
$<>"S" THEN N=N+1
180 IF A(D+1,E)=X THEN PRINT "WALL"
190 PRINT "SOUTH: ";
200 IF A(D-1,E)=S THEN PRINT "OPEN":IF A
$<>"N" THEN N=N+10
210 IF A(D-1,E)=X THEN PRINT "WALL"
220 PRINT "EAST: ";

```

```

230 IF A(D,E+1)=S THEN PRINT "OPEN":IF A
$<>"W" THEN N=N+100
240 IF A(D,E+1)=X THEN PRINT "WALL"
250 PRINT "WEST: ";
260 IF A(D,E-1)=S THEN PRINT "OPEN":IF A
$<>"E" THEN N=N+1000
270 IF A(D,E-1)=X THEN PRINT "WALL"
280 PEN 3:PAPER 1:LOCATE 1,19:PRINT " ME
MNONITE INDICATOR READS";:PRINT 100*SQR(
(Z-D)*(Z-D))+SQR((Y-E)*(Y-E))" ":PAPER
0:PEN 1
290 REM -----
300 GOSUB 1210:REM PRINTOUT
310 IF INKEY$<>" " THEN 310
320 W$=INKEY$: IF W$<>CHR$(240) AND W$<>C
HR$(241) AND W$<>CHR$(242) AND W$<>CHR$(
243) AND W$<>"H" AND W$<>"h" THEN 320
330 W$=UPPER$(W$):SOUND 1,(100+RND*100)
340 IF W$="H" THEN FF=-99:GOSUB 500:GOTO
440
350 IF W$<>"H" THEN T$=A$:GOSUB 1360
360 IF A$="N" AND A(D+1,E)=X THEN 320
370 IF A$="S" AND A(D-1,E)=X THEN 320
380 IF A$="E" AND A(D,E+1)=X THEN 320
390 IF A$="W" AND A(D,E-1)=X THEN 320
400 IF A$="N" THEN D=D+1
410 IF A$="S" THEN D=D-1
420 IF A$="E" THEN E=E+1
430 IF A$="W" THEN E=E-1
440 B(D,E)=46

```

```

450 IF Z=D AND Y=E THEN 1430
460 IF ABS(Z-D)<3 THEN IF ABS(Y-E)<3 THEN LOCATE 1,19:PRINT "You are very, very
close":SOUND 1,99,99,7,0,0,
1
470 WEND
480 END
490 REM -----
500 REM HELP
510 A(D,E)=42
520 CLS
530 REM 'HELP' PRINTOUT
540 PEN 2:PRINT "You are facing ";Q$:PRINT
550 PEN 1:PRINT TAB(13);"North":PRINT
560 PEN 3
570 FOR B=15 TO 1 STEP-1
580 PRINT TAB(9);
590 FOR C=1 TO 15
600 IF A(B,C)=42 THEN PEN 1:PRINT CHR$(2
24);:PEN 3:GOTO 630
610 IF B(B,C)=46 THEN PEN 2:PRINT "."::P
EN 3:GOTO 630
620 PRINT CHR$(A(B,C));
630 NEXT C
640 PRINT
650 NEXT B
660 PEN 1:PRINT:PRINT TAB(13);"South"
670 M=M+5
680 FOR DELAY=1 TO 1000:NEXT DELAY

```

```

690 CLS
700 GOSUB 1030
710 A(D,E)=S
720 RETURN
730 REM -----
740 REM INITIALISE
750 BORDER 0:MODE 1:INK 0,0:INK 1,26:INK
    2,18:INK 3,6:PEN 1:PAPER 0
760 PRINT "Please stand by while I build
    the maze"
770 RANDOMIZE TIME
780 DIM A(15,15),B(15,15)
790 B=INT(RND*3)
800 Z=13:Y=13
810 IF B=1 THEN Y=3
820 IF B=2 THEN Z=2
830 X=143:REM SOLID WALL 'BLOCK'
840 S=32:REM EMPTY PASSAGE
850 FOR B=1 TO 15
860 FOR C=1 TO 15
870 A(B,C)=X
880 IF C>1 AND B>1 AND C<15 AND B<15 THE
    N IF RND>0.9 THEN A(B,C)=S
890 NEXT C
900 NEXT B
910 D=2:E=2
920 CHECKSUM=0
930 FOR F=1 TO 69
940 READ B:READ C
950 CHECKSUM=CHECKSUM+B+C

```



```

960 A(B,C)=S
970 NEXT F
980 IF CHECKSUM<>1100 THEN PRINT "ERROR
IN DATA":END
990 WINDOW £1,16,27,2,8
1000 M=-5
1010 A$="E":T$="E":Q$="East":REM START
      FACING EAST
1020 FF=-99:RETURN
1030 PEN 2:LOCATE 1,1
1040 PRINT TAB(15);CHR$(222);:PEN 0:PAPE
R 1:PRINT "MAZE-O-SCOPE";:PAPER 0:PEN 2:
PRINT CHR$(223)
1050 FOR J=1 TO 7:PRINT TAB(15);CHR$(127
);:PRINT STRING$(12,32);:PRINT CHR$(127)
:NEXT J
1060 PRINT TAB(15);CHR$(221);STRING$(12,
127);CHR$(220)
1070 PEN 1
1080 RETURN
1090 REM -----
1100 DATA 2,2,2,3,2,4,2,5,2,6,2,7
1110 DATA 3,7,4,7,5,7,5,6,5,5,5,4,5,3,6,
3
1120 DATA 7,3,7,4,7,5,7,6,7,7,7,8,7,9,9,
8
1130 DATA 9,9,10,8,10,7,10,6,10,5,10,4
1140 DATA 10,3,11,3,12,3,13,3,14,3,14,2,
7,10
1150 DATA 6,10,5,10,4,10,3,10,2,10,2,11,

```

```

2,12
1160 DATA 2,13,2,14,6,11,6,12,6,13,6,13,
7,12
1170 DATA 14,12,8,12,8,14,9,12,9,13,9,14
,10,12
1180 DATA 11,9,11,10,11,11,11,12,12,9,13
,9,13,10
1190 DATA 13,11,13,12,13,13,13,4,14,14,1
0,9
1200 REM -----
1210 IF N=0 THEN P=8
1220 IF A$="H" THEN A$=T$
1230 IF A$="N" AND N=1 OR A$="S" AND N=1
0 OR A$="E" AND N=100 OR A$="W" AND N=10
00 THEN P=5
1240 IF A$="N" AND N=101 OR A$="E" AND N
=110 OR A$="W" AND N=1001 THEN P=1
1250 IF A$="N" AND N=100 OR A$="S" AND N
=1000 OR A$="E" AND N=10 OR A$="W" AND N
=1 THEN P=2
1260 IF A$="N" AND N=1000 OR A$="S" AND
N=100 OR A$="E" AND N=1 OR A$="W" AND N=
10 THEN P=3
1270 IF (A$="N" OR A$="S") AND N=1100 OR
(A$="E" OR A$="W") AND N=11 THEN P=4
1280 IF A$="N" AND N=1101 OR A$="S" AND
N=1110 OR A$="E" AND N=111 OR A$="W" AND
N=1011 THEN P=6
1290 IF A$="N" AND N=1001 OR A$="S" AND
N=110 OR A$="E" AND N=101 OR A$="W" AND

```

```

N=1010 THEN P=7
1300 ORIGIN 250,270:MOVE 0,0
1310 IF FF=P THEN RETURN
1320 FF=P
1330 PAPER £1,3:CLS £1,
1340 ON P GOSUB 1500,1550,1600,1650,1700
,1750,1800,1860
1350 REM -----
1360 IF W$=CHR$(240) OR A$="H" THEN A$=T
$
1370 IF W$=CHR$(241) AND T$="N" OR W$=CH
R$(243) AND T$="E" OR W$=CHR$(242) AND T
$="W" THEN A$="S"
1380 IF W$=CHR$(241) AND T$="S" OR W$=CH
R$(243) AND T$="W" OR W$=CHR$(242) AND T
$="E" THEN A$="N"
1390 IF W$=CHR$(243) AND T$="N" OR W$=CH
R$(242) AND T$="S" OR W$=CHR$(241) AND T
$="W" THEN A$="E"
1400 IF W$=CHR$(242) AND T$="N" OR W$=CH
R$(243) AND T$="S" OR W$=CHR$(241) AND T
$="E" THEN A$="W"
1410 RETURN
1420 REM -----
1430 REM END OF GAME
1440 FOR J=1 TO 20
1450 PRINT £1,"YOU FOUND IT IN JUST" M MOV
ES.... "
1460 FOR H=1 TO 100:NEXT H
1470 SOUND 5,RND*90,RND*90

```

```

1480 NEXT J
1490 LOCATE 1,22:END
1500 REM P=1
1510 GOSUB 1910:REM LEFT CLOSED
1520 GOSUB 1940:REM FRONT OPEN
1530 GOSUB 2000:REM RIGHT OPEN
1540 RETURN
1550 REM P=2
1560 GOSUB 1910:REM LEFT CLOSED
1570 GOSUB 2070:REM FRONT CLOSED
1580 GOSUB 2000:REM RIGHT OPEN
1590 RETURN
1600 REM P=3
1610 GOSUB 2130:REM LEFT OPEN
1620 GOSUB 2070:REM FRONT CLOSED
1630 GOSUB 2190:REM RIGHT CLOSED
1640 RETURN
1650 REM P=4
1660 GOSUB 2130:REM LEFT OPEN
1670 GOSUB 2070:REM FRONT CLOSED
1680 GOSUB 2000:REM RIGHT OPEN
1690 RETURN
1700 REM P=5
1710 GOSUB 1910:REM LEFT CLOSED
1720 GOSUB 1940:REM FRONT OPEN
1730 GOSUB 2190:REM RIGHT CLOSED
1740 RETURN
1750 REM P=6
1760 GOSUB 2130:REM LEFT OPEN
1770 GOSUB 1940:REM FRONT OPEN

```

```

1780 GOSUB 2000:REM RIGHT OPEN
1790 RETURN
1800 REM P=7
1810 GOSUB 2130:REM LEFT OPEN
1820 GOSUB 1940:REM FRONT OPEN
1830 GOSUB 2190:REM RIGHT CLOSED
1840 RETURN
1850 RETURN
1860 REM P=8
1870 GOSUB 1910:REM LEFT CLOSED
1880 GOSUB 2070:REM FRONT CLOSED
1890 GOSUB 2190:REM RIGHT CLOSED
1900 RETURN
1910 REM LEFT CLOSED
1920 DRAW 0,110:DRAW 50,77:MOVE 50,34:DR
AW 0,0
1930 RETURN
1940 REM FRONT OPEN
1950 MOVE 50,34:DRAW 73,50:DRAW 73,62
1960 MOVE 73,50:DRAW 93,50:DRAW 93,62
1970 DRAW 73,62:DRAW 50,77:MOVE 128,26
1980 DRAW 93,50:DRAW 93,62:DRAW 128,84
1990 RETURN
2000 REM RIGHT OPEN
2010 MOVE 117,77
2020 DRAW 128,84:DRAW 128,26:DRAW 150,26
2030 DRAW 150,10:DRAW 166,0:DRAW 166,110
2040 DRAW 150,100:DRAW 150,10:DRAW 150,8
4
2050 DRAW 128,84:MOVE 128,26:DRAW 117,34

```

```

2060 RETURN
2070 REM FRONT CLOSED
2080 MOVE 50,77
2090 DRAW 117,77: DRAW 117,34
2100 DRAW 50,34: DRAW 50,77
2110 MOVE 40,26: DRAW 50,34
2120 RETURN
2130 REM LEFT OPEN
2140 DRAW 16,10: DRAW 16,100: DRAW 0,110
2150 DRAW 16,100: DRAW 16,84: DRAW 40,84
2160 DRAW 40,26: DRAW 16,26: DRAW 40,26
2170 DRAW 50,34: MOVE 50,77: DRAW 40,84
2180 RETURN
2190 REM RIGHT CLOSED
2200 MOVE 117,34
2210 DRAW 166,0: DRAW 166,110: DRAW 117,77
2220 RETURN

```


INTO THE UNKNOWN

The title of this adventure is not far from the truth, because you never know – from run to run of the program – just what you will encounter next.

Your clever Amstrad chooses 40 locations from its database of 100, and then constructs a map around the chosen locations. The computer then randomly distributes the objects and hazards throughout the adventure environment. This ensures that the game will be different each time you tackle it.

Your task is to solve the adventure by finding ten pieces of treasure, at the same time making your way through the countryside until you reach the Citadel. This is where safety lies.


You control your direction with the letters N, S, E and W, and use T to take an object. There are other commands, but it is up to you to discover them!



```

10 REM*****INTO THE UNKNOWN*****
20 MODE 1:CLS
30 INK 2,1,26
40 PRINT "                INTO THE UNKNOWN"
50 PRINT:PRINT
60 GOSUB 1620:GOSUB 710
70 CLS
80 K=10:ST=90
90 REM*****MAIN LOOP****
100 ST=ST-0.1:IF ST<0 THEN PRINT "YOU'RE
    FINISHED!":END
110 BORDER INT (RND*26)
120 SOUND 2,(14*40)+40
130 PRINT:PRINT"\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\
\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\
140 PRINT "YOU ARE ";R$(P):PRINT
150 IF K<8 THEN 200
160 PRINT "EXITS ARE: ";:IF V(P,1)>0 THEN
N PRINT "NORTH: ";
170 IF V(P,2)>0 THEN PRINT "SOUTH: ";

```




```

180 IF V(P,3)>0 THEN PRINT "EAST: ";
190 IF V(P,4)>0 THEN PRINT "WEST"
200 PRINT:PRINT
210 FOR T=1 TO 20
220 IF B(T,1)=P THEN PRINT B$(T):PRINT
230 NEXT
240 DV=0
250 FOR T=1 TO 10:IF M(T,1)=P THEN PEN 2
:PRINT M$(T);" ATTACKS...":DV=1:PEN 1
260 NEXT
270 PRINT:PRINT
280 INPUT P$
290 IF DV=1 THEN GOTO 380
300 IF (P$="S" OR P$="E") AND (R$(P)="NE
XT TO A SNAKE-INFESTED PIT" OR R$(P)="NE
AR A CLIFF" OR R$(P)="NEAR A CLIFF EDGE"
OR R$(P)="IN THE CASTLE TURRET" OR R$(P
)="ALONG THE BATTLEMENTS") THEN PRINT "Y
OU FALL... THAT'S THE END":END
310 IF P$="N" AND RIGHT$(R$(P),4)="CAVE"
THEN PRINT "MADNESS, GOING NORTH TAKES
YOU FURTHER INTO THE COLD...YOU FREEZE
TO DEATH":END
320 IF P$="H" OR P$="J" AND ST>20 THEN S
T=ST-18:PRINT "YOU'VE JUMPED TO ANOTHER
PLACE":PRINT:PP=P:P=INT(RND*38)+1:GOTO 6
20
330 IF P$="N" AND V(P,1)<>0 THEN P=V(P,1
):GOTO 620
340 IF P$="S" AND V(P,2)<>0 THEN P=V(P,2

```

```

1:GOTO 620
350 IF P$="E" AND V(P,3)<>0 THEN P=V(P,3)
1:GOTO 620
360 IF P$="W" AND V(P,4)<>0 THEN P=V(P,4)
1:GOTO 620
370 IF DV<>1 THEN GOTO 500
380 REM****FIGHT OR RUN****
390 IF P$<>"F" AND P$<>"R" THEN PRINT "Y
OU CAN ONLY FIGHT OR RUN IN THIS      S
ITUATION":ST=ST-3:GOTO 270
400 IF P$="F" THEN 440
410 IF P$="R" AND RND<0.36 THEN PRINT "Y
OU MUST STAY AND FIGHT":ST=ST-7:GOTO 440
420 PRINT "OK. YOU HAVE RUN AWAY":ST=ST-
3:PP=P:P=INT(RND*38)+1:IF P=PP THEN P=P+
1
430 GOTO 90
440 REM*****FIGHT*****
450 IF RND<0.05 THEN PRINT "YOU ARE BADL
Y WOUNDED":ST=ST-(INT(ST/3)):PRINT
460 FOR T=1 TO 10:IF M(T,1)=P THEN FV=T:
T=10
470 NEXT:ST=ST-(INT(M(FV,2)*1.5)):IF ST>
0 THEN PRINT "YOU ARE THE VICTOR":M(FV,1)
)=50:SOUND 1,70,40:GOTO 90
480 PRINT "YOU ARE EXHAUSTED...":GOSUB 1
780:PRINT "YOU ARE DYING FAST...":GOSUB
1780:PRINT "YOU ARE DEAD":END
490 REM*****STRENGTH*****
500 OC=0:FOR T=1 TO 10:IF B(T,1)=0 THEN

```

```

OC=OC+1
510 NEXT
520 IF OC>5 AND (P$="STRENGTH" OR LEFT$(
P$,2)="SC") THEN PRINT:PRINT "STRENGTH I
S ";ST:GOTO 90
530 REM*****TAKE*****
540 OT=0:FOR T=1 TO 20
550 IF B(T,1)=P AND P$="T" THEN B(T,1)=0
:OT=T:T=10
560 NEXT
570 IF OT=0 AND P$="T" THEN PRINT "NOTHI
NG TO TAKE":GOTO 90
580 IF P$="T" AND B(OT,2)=1 THEN PRINT "
YOU HAVE TAKEN THE OBJECT/S":PRINT:GOTO
600
590 IF P$="T" AND B(OT,2)=2 THEN PRINT "
MORE TREASURE TO YOUR COLLECTION":PRINT
600 IF P$="T" THEN 620
610 PRINT:PRINT "YOU CANNOT DO THAT!":PR
INT:GOTO 90
620 PRINT:PRINT "O.K.":SOUND 1,INT(RND*3
50)+50
630 K=0
640 FOR T=11 TO 20
650 IF B(T,1)<>0 THEN K=K+1
660 NEXT T
670 IF K=0 AND P=40 THEN GOTO 1700
680 PRINT
690 PRINT " ";K;"TREASURES STILL TO COL
LECT"

```

```

700 GOTO 90
710 REM***INITIALISATION***
720 DIM V(40,4)
730 RANDOMIZE TIME
740 FOR T=1 TO 40
750 V(T,1)=0
760 V(T,2)=0
770 V(T,3)=0
780 V(T,4)=0
790 FOR Z=1 TO INT(RND*4)+1
800 A=INT(RND*4)+1
810 IF A=T THEN 800
820 V(T,A)=INT(RND*39)+1
830 NEXT Z
840 NEXT T
850 P=1
860 DIM R$(40):DIM SB$(20)
870 RESTORE
880 FOR T=1 TO 20:READ SB$(T):NEXT
890 DATA IN THE GARDEN OF AN OLD HOUSE, I
N AN OLD HOUSE, IN A STRANGE CAVE, NESTLIN
G IN A HOLLOW IN THE GROUND
900 DATA IN A QUIET CLEARING, NEXT TO A S
NAKE-INFESTED PIT, STANDING BY A HIGH FEN
CE, IN A SMALL ROOM
910 DATA IN A DIRTY BACK STREET, IN A GAR
AGE, IN A BARN, IN THE CASTLE TURRET
920 DATA BY AN ORCHARD, IN THE MARKET PLA
CE, BY A MARKET STALL, IN A QUIET STREET
930 DATA IN A SMALL CHAMBER, BY A RED-BRI

```

CKED WALL, IN THE MUSIC ROOM, IN A DARK TUNNEL

940 TL=0

950 FOR T=1 TO 80

960 IF RND<0.5 AND TL<40 THEN TL=TL+1:READ R\$(INT((T+1)/2))

970 NEXT

980 FOR T=1 TO 40: IF R\$(T)="" THEN R\$(T)=SB\$(INT(RND*20)+1)

990 NEXT T

1000 R\$(40)="IN THE CITADEL- A PLACE OF SANCTUARY"

1010 DATA IN A LAKE, BY A CLEAR LAGOON, ON A SANDY BEACH, ON A PEBBLE BEACH

1020 DATA IN A FAST-FLOWING RIVER, IN THE CALM SEA, ON A CLIFF EDGE, NEAR A CLIFF

1030 DATA IN A CLEARING, BY THE STREAM, CLOSE TO THE FOREST, IN THE FOREST

1040 DATA IN A WOOD, IN THE MIDDLE OF PASTURE, SURROUNDED BY GREEN FIELDS

1050 DATA IN THE DESERT, WITHIN SIGHT OF AN OASIS, ON A HILLTOP, BY SOME CROSSROADS

1060 DATA ALONG A PATHWAY, NEXT TO A PATH, BY A BEATEN TRACK, ON SOME WASTELAND

1070 DATA IN A WOODED COPSE, NEAR A WATER FALL, IN SOME BEAUTIFUL ORNAMENTAL GARDENS

1080 DATA NEXT TO A FLOWEBED, ON A PLATEAU OF GRASSLAND, SURROUNDED BY WEEDS

1090 DATA IN A LARGE CAVERN, IN THE DEPTH
S OF AN ICY CAVE, IN A CAVE FILLED WITH R
OCKY OUTCROPS

1100 DATA AT A HILL'S BASE, UP TO YOUR KN
EES IN MUD, CLOSE TO A RIVER BANK

1110 DATA ON A ROCKY PLAIN, IN A DESOLATE
AREA, IN A COLD AND FLAT SURROUND, ON A G
RASSLESS BANK, IN A SMALL DEPRESSION

1120 DATA BY A CASTLE WALL, IN THE CASTL
E GROUNDS, IN THE GIANT HALLWAY, IN THE KI
TCHENS

1130 DATA IN THE SPLENDED MASTER BEDROOM
, IN THE STUDY LINED WITH BOOKS

1140 DATA WITHIN THE MAGNIFICENT GARDENS
, IN A DANK STORE ROOM, IN A APARSE RECEPT
ION ROOM

1150 DATA IN A PASSAGEWAY, ALONG A CORRID
OR, BY A SMALL WELL, IN A PAVED QUADRANGLE

1160 DATA IN THE MAIN TOWER, ALONG THE BA
TTLEMENTS, HIDING IN THE TORTURE CHAMBER

1170 DATA ON THE DRAW BRIDGE, NEAR THE MO
AT, IN THE MOAT, IN THE DUNGEONS

1180 DATA IN THE DRAWING ROOM, IN A STUDI
O WITH PAINTINGS ON THE WALLS, IN AN EMPT
Y BEDROOM

1190 DATA IN A TINY HALLWAY, IN THE LIBRA
RY, OUT ON THE PORCH

1200 DATA IN A REMOTE HAMLET, IN A SMALL
UNTIDY HUT, IN A PIG STY, IN A SHEEP PEN

1210 DATA IN THE SHEEP DIP UGH!, BY THE H

EN HOUSE, IN THE HEN HOUSE, IN THE FARMYARD

1220 DATA BY A SIGN POST, AT SOME CROSSROADS, IN A COACH HOUSE, SURROUNDED BY HAYSTACKS

1230 DATA NEAR THE BOILER ROOM, IN THE REGENT BATHROOM

1240 RESTORE 1310

1250 DIM B\$(20), B(20,2)

1260 FOR T=1 TO 20

1270 READ B\$(T)

1280 B(T,1)=INT(RND*38)+1

1290 IF T<11 THEN B(T,2)=1 ELSE B(T,2)=2

1300 NEXT T

1310 DATA "A ROWING BOAT", "A GARDENING FORK", "SOME MATCHES", "A KNIFE", "A SPADE"

1320 DATA "A COPY OF THE COUNTRY CODE", "A GLASS OF WATER", "SOME ROPE"

1330 DATA "A PAIR OF BINOCULARS", "A MAGNIFYING GLASS"

1340 DATA A GOLD PENDANT, A 40 CARAT DIAMOND, A LARGE RUBY, A PLATINUM BANGLE, A JEWEL ENCRUSTED HEADBAND

1350 DATA A BAG OF COINS, A KRUGERAND, AN OLD MASTER, A SAPPHIRE, THE MAP TO CAPTAIN KIDD'S TREASURE

1360 DIM M\$(10), M(10,2), DM\$(20), DM(20,2)

1370 FOR T=1 TO 20: READ DM\$(T): READ DM(T,1): READ DM(T,2): NEXT T: DM(17,1)=INT(RND*38)+1

```

1380 FOR T=0 TO 9:IF RND>0.05 THEN K=10
ELSE K=1
1390 M$(T)=DM$(K+T):M(T,1)=DM(K+T,1):M(T
,2)=DM(K+T,2)
1400 NEXT T
1410 DATA A BLACK KNIGHT,10,10
1420 DATA A VENOMOUS SNAKE,4,5
1430 DATA A GIANT BEAR,20,12
1440 DATA A VIOLENT SERF,2,2
1450 DATA A DRUNKEN SWORDSMAN,16,4
1460 DATA A SWORDSMAN,32,8
1470 DATA A PACK OF WOLVES,37,7
1480 DATA A RABID DOG,21,5
1490 DATA A SCORPION,7,7
1500 DATA A GUARD DOG,9,5
1510 DATA A GREEDY TROLL,26,8
1520 DATA THE WIZARD OF ZOR,40,15
1530 DATA AN OLD WARLOCK,34,9
1540 DATA THE MIGHTY WARRIOR OF TROTH,39
,15
1550 DATA A MAD VIKING,15,12
1560 DATA THE BARON OUT HUNTING,28,8
1570 DATA THE MADBEAST,0,13
1580 DATA A PLAGUE OF GIANT RATS,24,6
1590 DATA A BAND OF THIEVES,31,11
1600 DATA A GROUP OF KILLER BATS,15,5
1610 RETURN
1620 REM*****INSTRUCTIONS*****
1630 PRINT "      You must attempt to coll
ect the ten treasures and take them to t

```


1800 RETURN


GOTHAM RUNNER

The world as we know it has ended. Despite the years of desperate talking, one day the big bombs started falling. In the aftermath, there were no victors, only survivors.

You are one of the survivors. Aided by a mutant pit pony you have befriended, you live a precarious life in the ruins of New York City. In *Gotham Runner*, you and your pony are trying to cross the 100-mile width of the city. A rabid pack of mutant dogs has got your scent, and is relentlessly pursuing you.

As you can tell from this morbid description, you have a mass of problems – and there are many more of them in store, as you'll discover when you run the adventure. There are several incidents which will occur from time to time (many of them will not occur during any particular run) and no incident will happen more than once in a run.

It is simple to play, as you only have to select an action from those presented to you on the screen. When you start running the program, you'll see something like this on your monitor:




NUMBER OF DAYS ELAPSED: 1
YOU HAVE TRAVELLED 32 MILES
THE DOG PACK HAS COVERED 8 MILES
IT IS 24 MILES BEHIND YOU

ENTER YOUR COMMAND:

- 1 - EAT FROM YOUR SUPPLIES
- 2 - PROCEED CAUTIOUSLY
- 3 - MAKE A RUN FOR IT
- 4 - TAKE A REST STOP

Despite the simplicity, however, you'll find it extremely difficult to get through the city, and you'll have to play it many many times to work out the optimum strategy.

When you're ready to run across the ruins, enter the following listing:



```

10 REM GOTHAM RUNNER
20 GOSUB 630:REM INITIALISE
30 M=M+1
40 B=B-1
50 IF B>0 THEN 110
60 PRINT "YOU SHOULD HAVE EATEN. YOU SLU
MP          DOWN AND DIE..."
70 PRINT:PRINT "THE DOG PACK SNIFFS THE
AIR...AND          ADVANCES RAPIDLY..."
80 GOSUB 1530:END
90 REM -----
100 REM MAIN CYCLE
110 CLS
120 PRINT
130 PRINT "NUMBER OF DAYS ELAPSED:"M
140 PRINT:PRINT "YOU HAVE TRAVELLED"T"ML
ES":PRINT
150 IF N=0 THEN 190
160 IF T-N<1 THEN PRINT "THE DOG PACK HA
S GOT YOU":GOSUB 1530:END
170 PRINT "THE DOG PACK HAS COVERED"N"MI
LES"
180 PRINT:PRINT "IT IS"T-N"MILES BEHIND
YOU"
190 PRINT:PRINT
200 IF RND>0.8 THEN GOSUB 710
210 IF D<2 THEN PRINT "YOU SHOULD LOOK F
OR NEW FOOD SUPPLIES":PRINT
220 PRINT:IF B<2 THEN PRINT "YOU ARE GET
TING VERY HUNGRY...":PRINT


```

```
230 WHILE INKEY$<>"":WEND
240 PRINT
250 PRINT "ENTER YOUR COMMAND:"
260 PRINT
270 PRINT TAB(8);"1 - EAT FROM YOUR SUPP
LIES"
280 PRINT TAB(8);"2 - PROCEED CAUTIOUSLY
"
290 PRINT TAB(8);"3 - MAKE A RUN FOR IT"
300 PRINT TAB(8);"4 - TAKE A REST STOP"
310 A$=INKEY$:WHILE A$<"1" OR A$>"4":GOT
O 310:WEND
320 Z=VAL(A$)
330 PRINT
340 ON Z GOTO 370,410,440,480
350 PRINT
360 REM -----
370 IF D=0 THEN PRINT:PRINT "YOU HAVE NO
FOOD LEFT":GOTO 250
380 B=5
390 D=D-1
400 GOTO 500
410 C=C+2
420 T=T+4+INT(RND*3)
430 GOTO 500
440 C=C+3
450 T=T+6+INT(RND*5)
460 REM -----
470 GOTO 500
480 C=0
```

```

490 PRINT "YOU AND YOUR MUTANT PIT PONY
SPEND      A FEW FITFUL HOURS IN SLEEP..
."
500 IF C>9 THEN PRINT "YOUR PONY DIES. Y
OU SHOULD HAVE      RESTED IT MORE.":
GOSUB 1530:END
510 IF INT(RND*12)<>6 THEN 550
520 PRINT:PRINT "YOU FIND A SUPPLY OF
      UNCONTAMINATED FOOD!"
530 C=C-5
540 D=3
550 N=N+5+INT(RND*4)
560 IF T<100 THEN 590
570 PRINT "YOU HAVE TRAVERSED THE RUINED
      CITY      AND ESCAPED!":GOSUB 1530
580 PRINT "WELL DONE!":GOSUB 1530:END
590 PRINT "-----"
      "
600 GOSUB 1530
610 GOTO 30
620 REM -----

```



```
630 REM INITIALISATION
640 MODE 1:INK 1,24:INK 2,18:INK 3,26
650 RANDOMIZE TIME
660 T=INT(RND*20)+21
670 M=0:C=0:N=0
680 K=0:B=5:D=3
690 RETURN
700 REM -----
710 REM INCIDENTS
720 K=K+1
730 X=INT(RND*6)+1
740 IF K=6 THEN RETURN
750 IF H(X)=1 THEN K=K+1:GOTO 730
760 H(X)=1
770 ON X GOSUB 830,910,970,1040,1150,126
0
780 GOSUB 1530
790 B=B-INT(RND*2)
800 PRINT
810 RETURN
820 REM -----
830 REM MUTANT HUMAN
840 PRINT "AS YOU STARE IN HORROR, A HUM
AN BEING"
850 PRINT "DRESSED IN ROTTEN RAGS STUMBL
ES TOWARDS"
860 PRINT "YOU...AND EMBRACES YOU!":GOS
UB 1530
870 PRINT " HE IS LETHALLY":REM NOTE
SPACE WITHIN QUOTE MARKS AT START
```

```

880 PRINT "CHARGED WITH RADIATION..."
890 RETURN
900 REM -----
910 REM STORM
920 PRINT "A MASSIVE WIND STORM SWEEPS D
OWN THE"
930 PRINT "AVENUE..YOU AND YOUR PONY DAS
H TO"
940 PRINT "ONE SIDE IN A VAIN BID TO FIN
D SHELTER"
950 RETURN
960 REM -----
970 REM BURIED
980 PRINT "THE RUBBLE SHIFTS BENEATH YOU
R FEET...":GOSUB 1530
990 PRINT "YOU FIND YOURSELF SLIPPING DO
WN...":GOSUB 1530
1000 PRINT "AND SLOWLY DRAG YOURSELF, GR
EATLY"
1010 PRINT "WEAKENED, BACK TO STREET LEV
EL..."
1020 RETURN
1030 REM -----
1040 REM SNAKE
1050 PRINT "YOU HEAR THE DREADFUL SOUND
OF A":GOSUB 1530
1060 PRINT "RATTLESNAKE...IT STRIKES..."
;:GOSUB 1530
1070 PRINT "BUT AT THE"
1080 PRINT "SAME TIME YOU SWING OUT AT I

```


T WITH"

1090 PRINT TAB(8); "YOUR CLUB...":GOSUB 1530

1100 PRINT:PRINT "IN THE CONFUSION, YOU DO NOT KNOW IF"

1110 PRINT "IT HAS BITTEN YOU, OR WHETHER YOU HIT"

1120 PRINT "IT BEFORE IT COULD STRIKE..."

1130 RETURN

1140 REM -----

1150 REM PHOSPHORUS ATTACK

1160 PRINT "YOU HEAR THE DREADED SCREAMING OF NINE":GOSUB 1530

1170 PRINT "CRAZED MEMBERS OF THE PHOSPHOR GANG...":GOSUB 1530

1180 PRINT "THEY ATTACK...":GOSUB 1530

1190 PRINT "AND AS YOU WHEEL AWAY ON"

1200 PRINT "YOUR LOYAL PIT PONY, THEY THROW A":GOSUB 1530



```

1210 PRINT "POWDERED WHITE PHOSPHORUS BO
MB AT YOU!":GOSUB 1530
1220 PRINT "IT EXPLODES WITH A FIERCE WH
ITE GLOW,":GOSUB 1530
1230 PRINT "AND FOR A MOMENT OR TWO YOU
ARE BLINDED"
1240 RETURN
1250 REM -----
1260 REM MUTANT PUMA
1270 PRINT "YOU HEAR A SNARL BEHIND YOU,
AND SEE":GOSUB 1530
1280 PRINT "A BLACK MUTANT PUMA READY TO
LEAP...":GOSUB 1530
1290 PRINT "YOU DIG YOUR HEELS INTO YOUR
PONY'S"
1300 PRINT "SIDES...":GOSUB 1530
1310 PRINT "AND ATTEMPT TO RUN AWAY FROM
"
1320 PRINT "IT...":GOSUB 1530
1330 PRINT "WILL YOU SUCCEED?"
1340 GOSUB 1530:GOSUB 1530
1350 IF RND>0.7 THEN 1380
1360 PRINT TAB(4);"YOU SEEM TO HAVE REAC
HED SAFETY!"
1370 RETURN
1380 PRINT "THE PUMA LEAPS...":GOSUB 15
30
1390 PRINT "AND YOUR PONY GOES"
1400 PRINT "DOWN...":GOSUB 1530

```

```
1410 PRINT "THE PUMA'S JAWS TIGHTEN AROU  
ND"  
1420 PRINT "YOUR LEGS";:GOSUB 1530  
1430 PRINT ", AS YOU KICK WILDLY...":GOS  
UB 1530  
1440 PRINT:PRINT  
1450 PRINT "YOU ";:GOSUB 1530  
1460 PRINT "KNOW ";:GOSUB 1530  
1470 PRINT "THIS ";:GOSUB 1530  
1480 PRINT "IS ";:GOSUB 1530  
1490 PRINT "THE ";:GOSUB 1530  
1500 PRINT "END ";:GOSUB 1530  
1510 PRINT  
1520 END  
1530 REM DELAY  
1540 NOW=TIME  
1550 WHILE TIME-NOW<490:WEND  
1560 FOR J=800 TO 2 STEP -(RND*20+47)  
1570 SOUND 2,1000-J,7  
1580 IF RND>0.4 THEN 1600  
1590 SOUND 5,J,7  
1600 NEXT J  
1610 PEN (INT(RND*3)+1)  
1620 RETURN
```

GOLDEN NUGGET


We've saved the best until last. This first-rate adventure, from the creative mind of Brian Beesley, is set in the days of the Californian Gold Rush.

You direct your character around the town of Deadwood, and throughout the badlands around town, looking for gold. Most commands can be entered in the "verb noun" format.

You might think it mean of us, but we're not going to give you any further help. Just play the game and watch it unfold. Brian has used some excellent programming techniques to provide you with an extremely tricky adventure. We can't see you solving it for a long time. However, we'll give you a few hints in the form of the world's worst poem. *Don't read it if you want to try and solve the adventure unaided.*

Unequipped, no gold you'll glean,
Hazards avoided, before it is seen.
The mule needs care, as does the river,
Indians and bears may bless a giver.

With help like that, you're probably no better off than you were at the beginning . . .



```

10 REM*****
20 REM*****GOLDEN NUGGET*****
30 REM*****BEESLEY/GIFFORD*****
40 REM*****
50 L=1:MO=100
60 GOSUB 3470:GOSUB 2660
70 N=0
80 MODE 1:CLS
90 PRINT "      PRESS ANY KEY TO START"
100 N=N+1
110 IF INKEY$="" THEN 100
120 RANDOMIZE N
130 IF INT(RND*2)=1 THEN OB(17,1)=-1:LO(
16,5)=0:GR=1
140 CLS:PRINT "I'm ";LO$(L):PRINT STRING
$(40,"-");
150 OK=0:GOSUB 2470:PRINT STRING$(40,"-")
);GOTO 180
160 PRINT "I don't understand ":Q$
170 REM
180 REM*****COMMAND COLLECTION*****
190 REM
200 DC=0:OC=0:INPUT "What Now ";Q$:IF Q$
="" THEN 200
210 SP=INSTR(1,Q$," "):IF SP THEN 230
220 GOTO 1740
230 V$=LEFT$(Q$,SP-1):N$=MID$(Q$,SP+1)
240 IF (V$<>"60" AND LEN(V$)<3) OR LEN(N
$)<3 THEN 160
250 FOR VC=1 TO 18:IF INSTR(1,VB$(VC),V$

```

```

) THEN 270
260 NEXT VC:PRINT "I don't understand ";
V$:GOTO 180
270 FOR DC=1 TO 4:IF INSTR(1,ID$(DC),N$)
THEN 310
280 NEXT DC:FOR OC=1 TO 46:IF INSTR(1,OB
$(OC),N$) THEN 310
290 NEXT OC:PRINT "I don't understand ";
N$:GOTO 180
300 REM
310 REM*****VERB SUBROUTINE*****
320 REM
330 ON VC GOSUB 370,450,450,720,840,1000
,1140,1190,1260,1310,1410,1480,1530,1560
,1620
340 IF VC>15 THEN ON VC-15 GOSUB 1690,10
00,1970
350 IF OK THEN 140
360 GOTO 180
370 REM*****GO*****
380 IF DC>4 THEN PRINT "I go by the sun
and the stars":RETURN
390 IF LO(L,DC)=-1 THEN PRINT "Nothing t
hat way so I came back":RETURN
400 IF LO(L,DC)=0 THEN PRINT "I can't go
that way":RETURN
410 IF LO(LO(L,DC),5)=0 THEN 440
420 CM=0:GOSUB 2010:IF CM THEN 440
430 IF NOT CM THEN RETURN
440 L=LO(L,DC):OK=1:RETURN

```

```
450 REM*****BUY/GET*****
460 IF VC=3 AND LEFT$(N$,3)="WAT" THEN 6
20
470 GOTO 510
480 IF LD=4 AND OC<>13 THEN PRINT,, "I c
an't carry anymore" ELSE 560
490 IF VC=2 THEN PRINT "Do you think I'm
a mule"
500 RETURN
510 IF OB(OC,2)<0 OR OC=18 THEN PRINT "T
hat's not possible":RETURN
520 IF VC=2 AND (OB(OC,1)=0 OR OB(OC,1)=
50) THEN 550
530 IF VC=3 AND OB(OC,1)=0 THEN 550
540 GOTO 480
550 PRINT "I've got that already":RETURN
560 IF OB(OC,1)=L OR OB(OC,1)=L+50 THEN
580
570 IF (OB(13,1)=L OR OB(13,1)=0) AND OB
(OC,1)=50 THEN 580 ELSE 2440
580 IF VC=2 THEN 670
590 IF OB(OC,2)>0 THEN PRINT "That's ste
aling and I'm no thief":RETURN
600 OB(OC,1)=0:IF OC<>13 THEN LD=LD+1:GO
TO 660
610 TE=0:GOTO 660
620 IF L=21 OR L=22 OR L=23 OR L=30 THEN
630 ELSE 650
630 IF OB(5,1)=0 THEN WA=1:GOTO 660
640 PRINT "I've nothing to put it in":RE
```

TURN

650 PRINT "I can't reach the water":RETURN

660 PRINT "Right. I've got that":RETURN

670 IF OB(OC,2)<1 THEN PRINT "That's not
for sale":RETURN

680 IF MO<OB(OC,2) THEN PRINT "Not enough
money":RETURN

690 PRINT "That's bought":OB(OC,1)=0:MO=
MO-OB(OC,2):OB(OC,2)=0

700 IF OC<>13 THEN LD=LD+1

710 RETURN

720 REM*****LOAD*****

730 IF OC<>13 THEN PRINT "I can only load
the mule":RETURN

740 IF OB(13,1)=0 OR OB(13,1)=L THEN 770

750 IF OB(13,1)<>0 AND OB(13,1)<>L THEN
PRINT "The mule's not here"

760 RETURN

770 PRINT:INPUT "Which item shall I load
";LD\$

780 IF LEN(LD\$)<3 THEN PRINT "I don't understand
";LD\$:RETURN

790 FOR OC=1 TO 16:IF INSTR(1,OB\$(OC),LD
\$) THEN 810

800 NEXT OC:PRINT "I don't understand ";
LD\$:RETURN

810 IF OB(OC,1)=0 THEN OB(OC,1)=50:LD=LD
-1:PRINT "That's loaded":RETURN

820 IF OB(OC,1)=50 THEN PRINT "That's all"


```
ready on the mule":RETURN
830 PRINT "I don't have ";LD$:RETURN
840 ' SHOOT
850 IF OB(1,1)<>0 THEN PRINT "I don't ha
ve a gun":RETURN
860 IF OC=16 THEN 880
870 IF OC<>16 THEN 920
880 PRINT:PRINT "That's only wounded it
and made it very mad"
890 GOSUB 2430:PRINT:PRINT,,"It's chargi
ng at me now"
900 GOSUB 2430
910 PRINT:PRINT "The bear has killed me,
now we'll never get rich":END
920 IF OC=17 THEN 940
930 IF OC<>17 THEN 950
940 PRINT:PRINT "That's not doing any go
od":RETURN
950 IF OC=18 THEN 970
960 IF OC<>18 THEN 990
970 PRINT "The snake is dead":OB(OC,2)=0
:OB$(OC)="A DEAD SNAKE":WA=0
980 RETURN
990 PRINT "What good will that do?":RETU
RN
1000 ' DROP/LEAVE
1010 IF OB(OC,1)<>0 THEN 2450
1020 IF OC=6 AND L=9 THEN PRINT "The bea
r eats the honey and falls asleep"
1030 IF OC<>6 OR L<>9 THEN 1060
```

```

1040 OB$(16)="A SLEEPING BEAR":OB$(6)="A
N EMPTY POT OF HONEY":OB(6,1)=L:LD=LD-1
1050 LO(13,5)=0:RETURN
1060 IF OC=11 AND VC=17 THEN 1640
1070 IF OC=11 THEN PRINT "The bottle sma
shed":OB(OC,2)=-1:OB$(11)="A SMASHED WHI
SKEY BOTTLE"
1080 IF OC=11 THEN OB(OC,1)=L:LD=LD-1:RE
TURN
1090 OB(OC,1)=L:PRINT "I've done that":I
F OC<>13 THEN LD=LD-1
1100 IF OC<>13 THEN RETURN
1110 IF L<17 THEN OB(13,1)=INT(RND*12)+6
1120 IF L>21 THEN OB(13,1)=INT(RND*11)+2
0
1130 RETURN
1140 ' BLOW UP
1150 IF OB(9,1)<>0 THEN PRINT "I don't h
ave the dynamite":RETURN
1160 CLS:LOCATE 17,12:PRINT "BOOM!!"
1170 PRINT:PRINT:PRINT "I never was any
good with dynamite"
1180 PRINT "Now I'm finished":END
1190 ' SELL
1200 IF L<>4 THEN 1230
1210 IF OB(14,1)=0 THEN PRINT "The clerk
accepts the nugget and hands me $1000.
"
1220 IF OB(14,1)=0 THEN PRINT "Your shar
e is $500, now you can buy thatDisc Driv

```

```
e you wanted!":END
1230 IF OB(OC,1)<>0 THEN PRINT "I don't
have the ";N$
1240 IF OB(OC,1)=0 THEN PRINT "No one's
buying"
1250 RETURN
1260 '   DIG
1270 IF OB(OC,1)<>0 THEN PRINT "I've not
hing to dig with":RETURN
1280 IF OC=32 AND L=33 THEN PRINT "That
was hard work, but the tunnel is   clear
now":DV=1
1290 IF DV=1 THEN OB(32,1)=-1:LD(34,5)=7
:DV=0:RETURN
1300 PRINT "I don't think that will help
":RETURN
1310 '   DRINK
1320 IF OC<21 OR OC>25 THEN 1350
1330 IF WA=0 THEN PRINT "I have no water
"
1340 IF WA<>9 THEN 1370:RETURN
1350 IF OB(OC,1)<>0 THEN 2450
1360 IF OC<10 OR OC>11 THEN PRINT "I can
't drink that":RETURN
1370 PRINT "That tasted good"
1380 IF OC=10 THEN OB$(10)="AN EMPTY BEE
R GLASS"
1390 IF OC=11 THEN OB$(11)="AN EMPTY WHI
SKEY BOTTLE":EW=1
1400 RETURN
```

```

1410 ' MOVE
1420 IF OB(OC,1)=L OR OB(OC,1)=L+50 THEN
1440:DV=1
1430 IF DV=1 THEN DV=0:GOTO 2440
1440 IF OB(OC,2)<0 THEN PRINT "You must
be joking":RETURN
1450 IF OC=18 THEN OB(14,1)=36 ELSE 1470
1460 LO(13,5)=2:OB(26,1)=14:LO(26,5)=0:L
O(25,5)=5:IF GR THEN OB(17,1)=16:LO(15,5
)=3
1470 PRINT "I've done that":RETURN
1480 ' CLIMB
1490 IF L<>35 THEN PRINT "There's nowher
e to climb":RETURN
1500 IF DC<5 THEN 370
1510 IF OB(15,1)=0 THEN PRINT "Which dir
ection?":RETURN
1520 PRINT "I need rope to do that":RETU
RN
1530 ' LIGHT
1540 IF OC=8 AND OB(OC,1)=0 THEN LL=1:PR
INT "I see the light":RETURN
1550 PRINT "I can't do that":RETURN
1560 ' BREAK
1570 IF OB(7,1)<>0 THEN PRINT "I've noth
ing to do that with":RETURN
1580 IF OB(OC,1)=L OR OB(OC,1)=L+50 THEN
1600:DV=1
1590 IF DV=1 THEN DV=0:GOTO 2440
1600 IF OC=30 THEN PRINT "I think I can

```

```
get through now":OB(OC,1)=-1:LO(31,5)=7:
RETURN
1610 GOTO 1300
1620 '   GIVE
1630 IF OB(OC,1)<>0 THEN 2450
1640 IF OC=11 AND (L=15 OR L=16) AND EW=
0 THEN PRINT "The Indians take the whisk
ey and ride   off":DV=1
1650 IF DV=1 THEN DV=0:LD=LD-1:OB(OC,1)=
-1:OB(17,1)=-1:LO(16,5)=0:LO(15,5)=0:RET
URN
1660 IF OC=11 AND EW=1 THEN PRINT "Who w
ants an empty bottle?":RETURN
1670 IF L=9 AND LO(13,5)=1 THEN PRINT "I
'm not getting that close to the bear":R
ETURN
1680 PRINT "No on wants ";N$:RETURN
1690 '   TETHER
1700 IF OC=13 AND OB(OC,1)=0 THEN OB(OC,
1)=L:TE=1:PRINT "That's done":RETURN
1710 IF OB(OC,1)<>0 THEN PRINT "I don't
have the ";N$:RETURN
1720 PRINT "I don't understand":RETURN
1730 '
1740 '   SINGLE WORDS
1750 '
1760 IF LEN(Q$)<3 THEN 160
1770 FOR SC=1 TO 8:IF INSTR(1,SW$(SC),Q$
) THEN 1790
1780 NEXT SC:GOTO 160
```



```

1790 IF SC<5 THEN VC=1:DC=SC:GOTO 330
1800 ON SC-4 GOSUB 1820,1860,1880,1970
1810 GOTO 350
1820 ' HELP
1830 IF OB(13,2)>0 THEN 1850
1840 IF OB(13,1)<>0 AND TE=0 THEN PRINT
"The mule wasn't tethered, he's wandered
off!":RETURN
1850 PRINT "You're helping me!!!":RETURN
1860 ' LOOK
1870 OK=1:RETURN
1880 ' INVENTORY
1890 CLS:PRINT "I HAVE:-":NS=1:FOR OC=1
TO 15:IF OB(OC,1)=0 THEN PRINT OB$(OC):N
S=0
1900 IF OB(OC,1)=50 AND OB(13,1)=0 THEN
PRINT OB$(OC);" on the mule":NS=0
1910 NEXT OC
1920 IF MO=0 AND NS THEN PRINT "Nothing"
:RETURN

```

```
1930 IF MD=0 THEN PRINT "No money":RETURN
1940 IF MD=100 THEN PRINT "1 Dollar"
1950 IF MD<100 THEN PRINT MD;"Cents"
1960 RETURN
1970 '    HIDE
1980 IF L=14 AND LO(13,5)=2 THEN PRINT "
They've passed by":LO(13,5)=0:OB(26,1)=-
1:RETURN
1990 PRINT "That's not going to help":RE
TURN
2000 '
2010 '    DIRECTION CONDITIONS
2020 '
2030 ON LO(LO(L,DC),5) GOTO 2040,2060,20
90,2110,2140,2170,2220,2270,2320,2340,23
90
2040 '    BEAR 1
2050 PRINT "I'm not going that way while
that bear's there":RETURN
2060 '    BANDITS 2
2070 PRINT "The bandits have captured me
"
2080 PRINT "They've stolen the gold nugg
et, now    we'll never be rich":END
2090 '    INDIANS 3
2100 PRINT "The indians won't let me pas
s":RETURN
2110 '    PROVISIONS 4
2120 IF OB(4,1)=0 OR OB(4,1)=50 THEN CM=
```

```

1:RETURN
2130 PRINT "I'm getting hungry. I'm not
going any further without food":RETURN
2140 ' WATER 5
2150 IF WA AND (OB(5,1)=0 OR OB(5,1)=50)
THEN CM=1:RETURN
2160 PRINT "It's hot in this desert":GOS
UB 2430
2170 PRINT:PRINT "I'm dying of thirst":G
OSUB 2430
2180 INK 0,0:CLS:PEN 3:LOCATE 14,12:PRIN
T "I'M DEAD":LOCATE 1,24:END
2190 ' SHUTTERING 6
2200 IF OB(30,1)=-1 THEN CM=1:RETURN
2210 PRINT "I can't get past the shutter
ing":RETURN
2220 ' LAMP
2230 IF LL AND OB(8,1)=0 THEN 2250
2240 PRINT "It's too dark to go that way
":RETURN
2250 IF L=28 THEN LO(31,5)=11:GOTO 2400
2260 CM=1:RETURN
2270 ' ROPE 8
2280 IF OB(15,1)<>0 THEN 2300
2290 IF L<>35 THEN CM=1:RETURN
2300 IF OB(15,1)=0 THEN 2230
2310 PRINT "My skeleton will join those
on the floor":GOSUB 2430:END
2320 ' EARTH 9
2330 PRINT "I can't get past the dirt":R

```


ETURN

2340 ' DROWNING 10

2350 IF L=19 OR L=20 THEN 2370

2360 CM=1:RETURN

2370 PRINT "I don't think I can make it
to the bank":GOSUB 2430

2380 PRINT "I think I'm done for":GOSUB
2430:END

2390 ' MULE INSIDE 11

2400 IF OB(13,1)=0 THEN PRINT "I can't t
ake the mule inside":RETURN

2410 IF L=28 THEN LO(31,5)=7

2420 CM=1:RETURN

2430 FOR N=1 TO 3000:NEXT:RETURN

2440 PRINT "There is no ";N\$:RETURN

2450 PRINT "I don't have ";N\$:RETURN

2460 '

2470 ' WHAT IS SEEN

2480 '

2490 IF L=4 THEN PRINT "The clerk says,
'Have you come to sell me gold? I will
pay"

2500 IF L=4 THEN PRINT "\$1000 for a larg
e nugget. Can you find one?":RETURN

2510 IF L=19 OR L=20 THEN PRINT "I can't
see a thing":RETURN

2520 IF L=32 OR (L=33 AND LO(34,5)=0) TH
EN PRINT "The lamp is very dim, I can't
see very far":RETURN

2530 IF L=3 THEN PRINT "I see:~":PRINT S

```

L$:GOTO 2570
2540 PRINT "I See:--":NS=1:FOR OC=1 TO 46
:IF OB(OC,1)=L THEN PRINT OB$(OC):NS=0
2550 NEXT OC:IF NS THEN PRINT "Nothing o
f interest":RETURN
2560 IF L<3 THEN RETURN
2570 ON L-2 GOTO 2580,2630,2590,2610:RET
URN
2580 PRINT "The storekeeper says, 'What
are you      wanting stranger?':RETURN
2590 PRINT "The bartender asks, 'What'll
you have      stranger? Beer is 4 cents"
2600 PRINT "and whiskey is 6 cents.":RET
URN
2610 PRINT "The blacksmith says, 'If you
're needing transport the horse will cos
t you 75"
2620 PRINT "cents but you can have the m
ule for only25 cents.":RETURN
2630 RETURN
2640 REM
2650 REM*****READ DATA BASE*****
2660 REM
2670 REM*****LOCATIONS*****
2680 RESTORE:DIM LO$(36),LO(36,5)
2690 FOR N=1 TO 36:READ LO$(N):FOR M=1 T
O 5:READ LO(N,M):NEXT M,N
2700 SL$="PICK - 10 CENTS      SPADE - 9 C
ENTS      AXE - 10 CENTS      HONEY - 5 C
ENTS      LAMP - 4 CENTS      DYNAMITE 7 C

```

ENTS PROVISIONS-9 CENTS WATER FLAS
K 14 CENTS GUN AND BULLETS - 20 CE
NTS"

2710 REM*****VERBS*****

2720 DIM VB\$(18):FOR N=1 TO 18:READ VB\$(
N):NEXT

2730 REM*****DIRECTIONS*****

2740 FOR N=1 TO 4:READ ID\$(N):NEXT

2750 REM*****SINGLE WORDS*****

2760 FOR N=1 TO 8:READ SW\$(N):NEXT

2770 REM*****OBJECTS*****

2780 DIM OB\$(46),OB(46,2):FOR N=1 TO 46:
READ OB\$(N):FOR M=1 TO 2:READ OB(N,M):NE
XT M,N

2790 RETURN

2800 REM*****DATA BASE*****

2810 REM*****LOCATIONS*****

2820 DATA in the main street of Deadwood
,2,3,-1,4,0

2830 DATA in the main street of Deadwood
,7,6,1,5,0

2840 DATA in the General Stores,0,0,0,1,
11

2850 DATA in the Assay Office,0,1,0,0,11

2860 DATA in the Saloon,0,2,0,0,11

2870 DATA in the Stables,0,0,0,2,0

2880 DATA on a dusty trail,8,-1,2,-1,0

2890 DATA on a dusty trail,-1,9,7,-1,0

2900 DATA in a dense forest,10,13,-1,8,0

2910 DATA on a winding trail,11,-1,9,-1,

0

2920 DATA in a canyon,12,0,10,0,0

2930 DATA in a blind canyon,0,0,11,0,0

2940 DATA in a dense forest,-1,14,-1,9,1

2950 DATA in a dense forest,15,-1,0,13,0

2960 DATA on an open plain,16,-1,14,-1,0

2970 DATA on an open plain,17,-1,15,-1,3

2980 DATA on the bank of a fast-flowing
river,20,21,16,18,10

2990 DATA in a river gorge,19,17,0,0,10

3000 DATA being battered to death on the
rocks,18,18,18,18,0

3010 DATA drowning in deep water,17,17,1
7,17,0

3020 DATA on the bank of a wide river,22
, -1, -1, 17, 0

3030 DATA in the river it's shallow enou
gh to ford,23,-1,21,20,0

3040 DATA on the bank of a wide river,24
, -1, 22, -1, 0

3050 DATA on a dusty trail,25,-1,23,-1,4

3060 DATA in a desert,26,-1,24,-1,0

3070 DATA in a desert,-1,27,25,-1,5


3080 DATA at the foot of the hills,-1,28
, -1, 26, 0

3090 DATA on the hillside,-1,31,29,27,0

3100 DATA on a narrow path,28,-1,30,-1,0

3110 DATA by a little stream,29,-1,-1,-1
, 0

3120 DATA in a tunnel,0,32,0,28,6



```
3130 DATA in a tunnel,0,35,0,31,8
3140 DATA in a tunnel,0,34,0,35,8
3150 DATA in a twisting tunnel,0,0,36,33
,9
3160 DATA at the bottom of a hole,0,33,0
,32,7
3170 DATA in a small cavern,34,0,0,0,7
3180 REM*****VERBS*****
3190 DATA GO,BUY,GET,LOAD,SHOOT,DROP,BLO
WUP,SELL,DIG,DRINK
3200 DATA MOVE,CLIMB,LIGHT,BREAK,GIVE,TE
THER,LEAVE,HIDE
3210 REM*****DIRECTIONS*****
3220 DATA NORTH,EAST,SOUTH,WEST
3230 REM*****SINGLE WORDS*****
3240 DATA NORTH,EAST,SOUTH,WEST,HELP,LOO
K,INVENTORY,HIDE
3250 REM*****OBJECTS*****
3260 DATA A GUN AND BULLETS,53,20,A PICK
,53,10,A SPADE,53,9
3270 DATA PROVISIONS,53,9,DRINKING FLASK
```

,53,14,HONEY,53,5
 3280 DATA AN AXE,53,5,A LAMP,53,4,DYNAMI
 TE,53,7
 3290 DATA A GLASS OF BEER,5,3,A BOTTLE O
 F WHISKEY,5,6
 3300 DATA A HORSE,6,75,AN OLD MULE,6,25,
 A LARGE GOLD NUGGET,-1,0
 3310 DATA A ROPE,12,0,A LARGE BAR TO THE
 EAST,9,-1
 3320 DATA A BAND OF INDIANS,15,-1,A HUGE
 RATTLESNAKE,36,-1
 3330 DATA NO WAY OUT BUT THE WAY I CAME,
 12,-1
 3340 DATA A RAGING TORRENT,18,-1,DEEP WA
 TER,17,-1
 3350 DATA SHALLOW WATER,21,0,SHALLOW WAT
 ER,22,0
 3360 DATA SHALLOW WATER,23,0,A TRICKLE O
 F WATER,30,0
 3370 DATA BANDITS THEY'VE NOT SEEN ME YE
 T,-1,-1
 3380 DATA SAND EVERYWHERE,25,-1,SAND EVE
 RYWHERE,26,-1
 3390 DATA A TUNNEL INTO THE HILLSIDE,28,
 -1,SHUTTERING BLOCKS THE ENTRANCE,28,-1
 3400 DATA THE TUNNEL ABOVE ME SKELETONS
 ON THE FLOOR,35,-1
 3410 DATA LOOSE DIRT BLOCKING THE WAY,33
 ,-1
 3420 DATA A NARROW PATH,28,-1,A GENERAL

```

STORES AND AN ASSAY OFFICE,1,-1
3430 DATA A SALOON AND A STABLE,2,-1,DEA
DWOOD,7,-1
3440 DATA A FOREST,8,-1,A FOREST,10,-1,A
CANYON,10,-1
3450 DATA HIGH CLIFFS,11,-1,A FOREST,15,
-1,A RIVER,16,-1
3460 DATA A STREAM,29,-1,TREES AND BUSHE
S,14,1,A RIVER,24,-1," ",3,1
3470 REM*****TITLE PAGE*****
3480 CLS:LOCATE 13,10:PEN 2:PRINT "GOLDE
N NUGGET":PEN 1
3490 PEN 3:LOCATE 8,12:PRINT "BY B.Beesl
ey & C.Gifford":PEN 1
3500 FOR T=1 TO 50
3510 X=INT(RND*38)+1:Y=INT(RND*23)+1
3520 IF Y>9 AND Y<13 THEN 3510
3530 LOCATE X,Y:PRINT CHR$(231):SOUND 1,
INT(RND*150)+50
3540 NEXT
3550 FOR T=1 TO 1000:NEXT T
3560 REM*****INSTRUCTIONS*****
3570 CLS:MODE 1
3580 PRINT "      I hear that a man can ge
t rich      finding GOLD."
3590 PRINT "      I'll need some help thou
gh, I knew  you would oblige...anything
we find,    we'll share fifty-fifty"
3600 PRINT "      Tell me what to do by us
ing:-      single word commands e.g. LO

```


OK, EAST"

3610 PRINT,,"two-word commands e.g. BUY
GUN, GO EAST"

3620 FOR T=600 TO 60 STEP-10

3630 SOUND 1,T:NEXT

3640 RETURN



WRITING YOUR OWN ADVENTURES

WRITING YOUR OWN ADVENTURES

If you really enjoy playing adventures then there is a good chance that you will enjoy writing your own adventures, based on your own plot, even more. Actually constructing your own adventure is not as difficult as it sounds and once you have written a couple of adventures you will appreciate how much fun it can be.

Firstly, let me distinguish the adventure game from other games and simulations. In my opinion, the adventure game has the following elements:

- 1) Locations.
- 2) Usually some form of logical connection between the locations (though the logic is sometimes far from perfect).
- 3) Some form of reward or aim to work for; escape, for instance, or the accumulation of treasure. Without some objective, an adventure would become pointless and boring.
- 4) Some form of obstacle hindering the progress of the adventurer.
- 5) Some element of discovery — find the wizard, the exit, how to make your character move, and so on.
- 6) Some form of command with which adventurers can direct themselves around the locations and thus interact with the adventure itself.

Before you write any of the actual program you should plan out a number of things. You must have a plot, and descriptions of the locations, characters, objects and the links between them all. In addition, I recommend that you work out all the obstructions, puzzles and connections between locations. This information will be a great help when you are writing and, more importantly, debugging your adventure.

Let's start with the plot of your adventure. This will involve a scenario and some form of objective(s), as well as some major hinderances. Plots can be easy to come by,

searching for gold, finding your way out of a maze, rescuing a trapped maiden. Thinking up a novel plot is much more difficult but it does make for a vastly improved game. It can get rather tiring sitting in front of a screen trying to accumulate more treasure from a similarly-named castle that is guarded by similar-type monsters. What your adventure needs is something new or at least a little less common. 'Crash!', the adventure set inside the computer, is an example of such a plot.

The actual objectives and the scenario, or playing environment, are closely connected. Once you have determined one, the other is simple to think of. If you choose an underwater scenario, then an obvious objective would be to find a sunken wreck. Similarly, if you choose escaping from a maze as your objective then a mythological scenario with gods and minotaur-type monsters immediately springs to mind. Below are a few other suggested scenarios and objectives:

Small Island	You have been shipwrecked and must find a way to get off the island.
Office Building	As a secret agent, you are searching for some top secret files.
The Human Body	As an antibody you must find the source of infection and fight it fast.
Mysterious House	You have gone on a mystery tour for some fun, things have turned dangerous and you and several others have been left here. What are the secrets in the house?
Houses of Parliament	Can you find the leaked government report before the media do? Your chance of a place in the Cabinet depends on your success in this mission.

A plague of rats that have left the sewers are resting in the Underground system. Can you eradicate them before they are strong enough to attack again?

Once you have the plot, it is not too hard thinking up names and descriptions of locations. Obviously, some will need to be very specific as they tie in closely with the plot; in the human body scenario, all the locations will obviously have to be parts of the body. In most adventures, however, there is also a need for some less specific, even though not so interesting, locations. These locations add to the realism of the adventure; for instance, in the Houses of Parliament scenario, going from the main chamber in the House of Commons straight into the bar or the library may lack realism, and after a few locations, you will start to run out of names. Locations such as corridors, small storerooms, a dusty alcove, a windswept quadrangle and a committee room may add to the adventure.

89

I will discuss characters in the same breath as objects since in adventure programming they are often treated in the same way. Your plot will obviously determine a number of the characters and the objects, but you may decide to add others either to create further obstructions or to aid the adventurer in his struggle to reach his goal. Although I can think of adventures where there are no characters, all adventures need objects of some kind. The objects and the characters are linked most strongly with the puzzles in the adventure, and this is the next part under discussion.

In an adventure, the plot and the puzzles are the two most important factors to be considered. Again, some of the puzzles will obviously come from the plot, but almost certainly you will need a number of extra puzzles to complete the adventure. One can categorise puzzles into three types. The first type of puzzle is directly linked to the plot and must be solved before the adventure can be finished. The second type of puzzle is not directly linked to the plot but still needs to be solved for the adventure to be completed. The final type of puzzle need not be solved; but while it may add to the realism and the fun of the adventure, it may also slow down the adventurer who is not aware that it is unimportant. Below are examples of the first two types of puzzle:

Need To Be Solved

- 1) The castle is dark without the torch, you can still move around but each location is not described.
- 2) You need a gun if you want to get past the guard at the door. You must haggle with the trader in the village.
- 3) You have broken a vase in the office and, to get out without it being noticed, you must find some glue.
- 4) The door is not locked but it is too heavy for you to push open. It is huge and wooden.
- 5) You come to a large canyon, you cannot climb down and behind you is a herd of stampeding cattle. All you have is unlimited amounts of string. What are you going to do?

Those of you who enjoy lateral thinking puzzles should be able to solve the last problem. The solution is to fill the canyon with string, walk over the canyon — which is now full — and then quickly take out all the string.

You only need a couple of very difficult problems in an adventure but these should be supplemented with a number of 'red herrings'. Some suggested ones are indicated below:

- 1) You are carrying a torch, and you spot a toy chemical set; if the player examines or removes the set then the torch ignites it and the adventurer is blown to smithereens.
- 2) Is the rubber duck of any use? After playing whacky adventure games like Pimania no one can be sure.
- 3) A crimson fish (this one is too obvious for words!)

There are many quite simple puzzles which can be implemented in your own adventure: locked doors, creatures barring your way, I'm sure you know the sorts of puzzles.

You will need to make a 'map' of your adventure showing all the links between locations, positions of objects and so on. The majority of adventures use the compass points, N, S, E and W, for direction of movement, with maybe Up and Down. Your adventure can use these, as well as the four points — NE, NW, SE, SW — or it can use a different system. You will also need to formulate a program routine that transfers your position from one location to another. There are a number of different methods, from simple grid systems to complex systems that allow movement in 16 different directions (including through time!).

Once you have all this neatly down on paper, you can start the actual programming. As with all games, there are certain points to note. Make sure the screen is presented neatly with no words cut in two by the end of a line. Add a bit of sound and some graphics here and there but try to make them relevant to the program; that is, when someone fires a gun, have a sound indicating this, and when the adventurer has just killed a monster, turn the screen red. Small touches like these increase an

adventure game's interest. Graphic pictures of locations are fine as long as they appear quickly; there is nothing more tiresome than having to wait for ages while a screen fills with a drawing. The user will grow tired of the pictures very rapidly despite their quality.

Finally, a word on structuring your adventure program. Why bother to structure programs at all? There are a number of good reasons for doing so. Since BASIC is a poorly structured language — compared to a language such as FORTH — the programmer must add his own framework. Structuring a program speeds a program up, as it does not have to pass through irrelevant masses of code every loop. Structuring also eradicates duplication of program functions and helps to save much memory space that can be put to good use elsewhere.

Another reason, particularly applicable to adventure programmers, is that after leaving a program for a few days, you will start to forget where each routine is. By structuring a program, the adventure writer knows where each routine is and what it does, and this makes debugging a great deal easier.

After all this, I hope you will see that there is a lot of work to be done when writing an adventure. It is, however, within the bounds of most computer users. It is an ideal way to start programming once the beginner has written a few simple programs. Adventures, in short, whether you write them or play them, can be enormously enjoyable.

BIBLIOGRAPHY

BIBLIOGRAPHY

This area of computing has not been overflowed with publications, and only recently have a number of books come onto the market — some excellent, some awful. Below I mention only the good ones.

While a number of these books may not have been written specifically for your computer, the ones mentioned either have an informative text or adventures that can be easily converted to your machine.

Creating Adventure Games on Your BBC Micro

Ian Watt. Interface/Addison Wesley

Ian really does know his stuff when it comes to writing adventure programs. He has his own style of adventure writing and, in this book, he reveals all. It is a slim volume containing three adventures all in text so they can be converted to other computers.

Creating Adventure Programs on Your Computer

Andrew Nelson. Interface

I have met Andrew and he is full of interesting ideas, which he kindly shared with me. This book contains a number of adventures, all written in Microsoft BASIC — and easily converted to another computer. A particularly intriguing title is 'The Aftermath of the Asmorian Disaster'.

Adventure Writing

Aardvark-80, 2352 S. Commerce, Walled Lake, MI 48088, USA

This 16-page booklet, sold in the United States (for the exorbitant sum of \$5), is a terrific help to all adventure writers. The adventure program included, 'Death Ship', is broken down in detail and comes with an addendum offering versions for most home computers.

Creating Adventure Games on Your Dragon 32

Clive Gifford. Interface

Five full adventures are explained in detail, one of which has now been transferred to cassette and disc software. Three of the five adventures can be converted for use on other computers without much difficulty.

The ZX81 Pocket Book

Trevor Toms. Phipps Associates

Only one section is devoted to adventures but, in it, the author details a vastly different approach to adventure writing. The book may be worth buying, particularly if you can find it at a discounted price.

Creating Adventures on Your Spectrum

Peter Shaw/James Mortleman. Interface

I know Peter well — he is on the editorial board of *Your Spectrum* magazine — and this is a strong book with many novel adventures, some featuring excellent graphics. Another point of note: the illustrations in this book were drawn by Peter himself.

Writing BASIC Adventure Games for the TRS-80

Frank Dacosta. Tab Books

This is a most useful guide to writing adventures. It was the book I first cut my teeth on and it is still proving invaluable now.

Write Your Own Adventure Programs for Your Microcomputer

Tyler/Howarth. Osborne

At just under £2 this must be the best value adventure book around. Do not be put off by the childish presentation, for the book has some serious things to say.

MYSTERIOUS ADVENTURES FOR YOUR AMSTRAD

Now you can enter six mysterious worlds. Worlds which will tax your strength, your cunning, your resolve and your wits. Six mysterious adventures to pit you and your Amstrad against strange worlds of the imagination.

THE EXPLORER puts you down in medieval times, as you discover much of your knowledge of the modern world is of little use in fighting off disaster. Move from there INTO THE UNKNOWN where tricky programming ensures that – although the worlds created are coherent, and can be mapped – no two adventures will ever be the same.

In THE PATRIARCH every adventure cliché you can imagine (including dragons, treasure and a magic amulet) come together to give you a challenge, and a laugh or two. In 3-D MAZE, you're searching for the Philosopher's Stone, aided by a full graphic representation of the maze ahead of you.

GOLDEN NUGGET is a breath of fresh air for those who would be wealthy, as your brains and brawn come under stress in the Californian gold rush. If your reflexes are good, and your eye is keen, you'll survive a nightmare crossing of a post-Holocaust New York City in GOTHAM RUNNER, aided only by your 'mutant pit pony'.

The book includes a detailed explanation of how you can create your own completely original Adventures; and there is a reading list to help you develop even further your writing and playing skills.



MASTERS OF THE UNIVERSITY OF CALIFORNIA

AMSTRAD CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>